Vzájemná poloha přímky, roviny a kvadriky

```
> restart;
> with(LinearAlgebra):
> X:=Vector[row]([x,y,z,1]);
```

$$X := [x, y, z, 1]$$

Matice kvadriky:

```
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

Rovnice kvadriky:

```
> Kv:=expand(evalm(X&*K&*Transpose(X)))=0;
```

$$Kv := x^2\, a(1,1) + 2\, x\, y\, a(1,2) + 2\, x\, z\, a(1,3) + 2\, x\, a(1,4) + y^2\, a(2,2) + 2\, y\, z\, a(2,3) + 2\, y\, a(2,4) + z^2\, a(3,3) + 2\, z\, a(3,4) + a(4,4) = 0$$

Vzájemná poloha přímky a kvadriky

Bod přímky:

```
> M:=[m,n,p];
```

$$M := [m, n, p]$$

Parametrické rovnice přímky:

```
> Primka:=[x=m+t*u,y=n+t*v,z=p+t*w];
```

$$Primka := [x = m + t\, u, y = n + t\, v, z = p + t\, w]$$

Parametrické rovnice přímky dosazené do rovnice kvadriky:

```
> Kv1:=simplify(eval(Kv,Primka));
```

$$Kv1 := 2\, a(1,1)\, m\, t\, u + 2\, a(1,2)\, m\, t\, v + 2\, a(1,2)\, t\, u\, n + 2\, a(1,2)\, t^2\, u\, v + 2\, a(1,3)\, m\, t\, w + 2\, a(1,3)\, t\, u\, p + 2\, a(1,3)\, t^2\, u\, w + 2\, a(2,2)\, n\, t\, v + 2\, a(2,3)\, n\, t\, w + 2\, a(2,3)\, t\, v\, p + 2\, a(2,3)\, t^2\, v\, w + 2\, a(3,3)\, p\, t\, w + a(1,1)\, m^2 + 2\, a(1,4)\, m + a(2,2)\, n^2 + 2\, a(2,4)\, n + a(3,3)\, p^2 + 2\, a(3,4)\, p + a(1,1)\, t^2\, u^2 + 2\, a(1,2)\, m\, n + 2\, a(1,3)\, m\, p + 2\, a(1,4)\, t\, u + a(2,2)\, t^2\, v^2 + 2\, a(2,3)\, n\, p + 2\, a(2,4)\, t\, v + a(3,3)\, t^2\, w^2 + 2\, a(3,4)\, t\, w + a(4,4) = 0$$

Koeficienty A, B, C rovnice $At^2 + Bt + C = 0$ společných bodů kvadriky a přímky:

```
> A:=coeff(lhs(Kv1),t^2);
```

$$A := 2\, a(1,2)\, u\, v + 2\, a(1,3)\, u\, w + 2\, a(2,3)\, v\, w + a(1,1)\, u^2 + a(2,2)\, v^2 + a(3,3)\, w^2$$

```
> B:=coeff(lhs(Kv1),t);
```

$$B := 2\, a(1,1)\, m\, u + 2\, a(1,2)\, m\, v + 2\, a(1,2)\, u\, n + 2\, a(1,3)\, m\, w + 2\, a(1,3)\, u\, p$$

$$+ 2\,a(2,2)\,n\,v + 2\,a(2,3)\,n\,w + 2\,a(2,3)\,v\,p + 2\,a(3,3)\,p\,w + 2\,a(1,4)\,u + 2\,a(2,4)\,v + 2\,a(3,4)\,w$$

```
> C:=sort(coeff(lhs(Kv1),t,0));
```

$$C := a(1,1)\,m^2 + 2\,a(1,2)\,m\,n + 2\,a(1,3)\,m\,p + a(2,2)\,n^2 + 2\,a(2,3)\,n\,p + a(3,3)\,p^2 + 2\,a(1,4)\,m + 2\,a(2,4)\,n + 2\,a(3,4)\,p + a(4,4)$$

Vzájemná poloha roviny a kvadriky

Uvažujeme rovinu o rovnici $x = 0$

```
> KvR:=subs(z=0,Kv);
```

$$KvR := x^2\,a(1,1) + 2\,x\,y\,a(1,2) + 2\,x\,a(1,4) + y^2\,a(2,2) + 2\,y\,a(2,4) + a(4,4) = 0$$

1) $a(1,1) = 0$**,** $a(1,2) = 0$**,** $a(2,2) = 0$**,** $a(1,4) \neq 0$**,** $a(2,4) \neq 0$

```
> a(1,1):=0; a(1,2):=0; a(2,2):=0; a(1,4):=1; a(2,4):=2;
```

$$a(1,1) := 0$$
$$a(1,2) := 0$$
$$a(2,2) := 0$$
$$a(1,4) := 1$$
$$a(2,4) := 2$$

Rovnice kvadriky:

```
> K1:=K;
```

$$K1 := \begin{bmatrix} 0 & 0 & a(1,3) & 1 \\ 0 & 0 & a(2,3) & 2 \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ 1 & 2 & a(3,4) & a(4,4) \end{bmatrix}$$

```
> Kv1R:=KvR;
```

$$Kv1R := 2\,x + 4\,y + a(4,4) = 0$$

```
> a(4,4):=-2;
```

$$a(4,4) := -2$$

Průnik kvadriky a roviny $z = 0$

```
> Kv1R;
```

$$2\,x + 4\,y - 2 = 0$$

```
> Kv1:=eval(Kv,{a(1,3)=1,a(2,3)=1,a(3,3)=1,a(3,4)=1,a(4,4)=-2});
```

$$Kv1 := -2 + 2\,x\,z + 2\,x + 2\,y\,z + 4\,y + z^2 + 2\,z = 0$$

Graficke znázornění plochy a jejího průniku s uvažovanou rovinou:

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> Prunik:=plot3d([t,solve(eval(Kv1R,x=t),y),0],t=-6..6,s=-5..5,thi
  ckness=6,color=red):
```

```
> plots[implicitplot3d](Kv1,x=-5..5,y=-5..5,z=-5..5,axes=frame,grid=[30,30,30],style=patchcontour,color=grey,scaling=constrained,lightmodel=light1):
> Rov_z:=plot3d(0,x=-8..8,y=-8..8,color=grey, style=wireframe):
> Plocha:=plots[implicitplot3d](Kv1,x=-5..5,y=-5..5,z=-5..5,axes=frame,color=COLOR(RGB,205/255,205/255,205/255),style=patchcontour,grid=[30,30,40],light=[90,-5,1,1,1],tickmarks=[3,3,3],orientation=[106,75],scaling=constrained):
> plots[display](Plocha,Prunik,Rov_z,scaling=constrained);
```




```
> a(4,4):='a(4,4)'; z:='z';
```

$$a(4, 4) := a(4, 4)$$

$$z := z$$

2) $a(1, 1) = 0$**,** $a(1, 2) = 0$**,** $a(2, 2) = 0$**,** $a(1, 4) = 0$**,** $a(2, 4) = 0$, $a(4, 4) = 0$

```
> a(1,1):=0; a(1,2):=0; a(2,2):=0; a(1,4):=0; a(2,4):=0;a(4,4):=0;
```

$$a(1, 1) := 0$$

$$a(1, 2) := 0$$

$$a(2, 2) := 0$$

$$a(1, 4) := 0$$

$$a(2, 4) := 0$$

$$a(4, 4) := 0$$

Rovnice kvadriky:

```
> K2:=K;
```

$$K2 := \begin{bmatrix} 0 & 0 & a(1,3) & 0 \\ 0 & 0 & a(2,3) & 0 \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ 0 & 0 & a(3,4) & 0 \end{bmatrix}$$

```
> Kv2:=factor(Kv);
```

$$Kv2 := z\,(a(3,3)\,z + 2\,x\,a(1,3) + 2\,a(2,3)\,y + 2\,a(3,4)) = 0$$

Průnik kvadriky a roviny $z = 0$

```
> Kv2R:=KvR;
```

$$Kv2R := 0 = 0$$

```
> Kv2:=eval(Kv,{a(1,3)=1,a(2,3)=1,a(3,3)=1,a(3,4)=1,a(4,4)=0});
```

$$Kv2 := 2\,x\,z + 2\,y\,z + z^2 + 2\,z = 0$$

```
> KvP:=solve(Kv2,z);
```

$$KvP := 0, -2\,x - 2\,y - 2$$

Graficke znázornění plochy, tj. dvou rovin, a jejího průniku s uvažovanou rovinou, kterým je jedna z těch rovin:

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> Rov1:=plot3d(0,x=-5..5,y=-5..5,axes=frame,color=COLOR(RGB,180/25
  5,180/255,180/255),style=patchnogrid,grid=[20,20],scaling=constr
  ained):
> Rov2:=plots[implicitplot3d]([op(2,factor(lhs(Kv2)))],x=-5..5,y=-
  9..7,z=-5..5,axes=frame,color=COLOR(RGB,153/255,153/255,153/255)
  ,style=patchcontour,grid=[20,20,20],scaling=constrained):
> plots[display](Rov1,Rov2,Rov_z,scaling=constrained,orientation=[
  120,70],tickmarks=[4,4,4]);
```



3) $a(1,1)=0$, $a(1,2)=0$, $a(2,2)=0$, $a(1,4)=0$, $a(2,4)=0$, $a(4,4)\neq 0$

```
> a(1,1):=0; a(1,2):=0; a(2,2):=0; a(1,4):=0;
  a(2,4):=0;a(4,4):=-10;
```

$$a(1,1) := 0$$
$$a(1,2) := 0$$
$$a(2,2) := 0$$
$$a(1,4) := 0$$
$$a(2,4) := 0$$
$$a(4,4) := -10$$

Rovnice kvadriky:

```
> K3:=K;
```

$$K3 := \begin{bmatrix} 0 & 0 & a(1,3) & 0 \\ 0 & 0 & a(2,3) & 0 \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ 0 & 0 & a(3,4) & -10 \end{bmatrix}$$

Průnik kvadriky a roviny $z = 0$

```
> Kv3R:=KvR;
```

$$Kv3R := -10 = 0$$

```
> Kv3:=eval(Kv,{a(1,3)=1,a(2,3)=2,a(3,3)=-1,a(3,4)=1,a(4,4)=-10});
```

$$Kv3 := -10 + 2\,x\,z + 4\,y\,z - z^2 + 2\,z = 0$$

Grafické znázornění uvažované plochy:

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> Plocha:=plots[implicitplot3d](Kv3,x=-15..15,y=-15..15,z=-15..15,
  axes=frame,orientation=[-50,70],color=COLOR(RGB,205/255,205/255,
  205/255),style=patchcontour,grid=[30,30,40],light=[90,-5,1,1,1],
  tickmarks=[3,3,3],scaling=constrained):
> Rov_z:=plot3d(0,x=-20..20,y=-20..20,color=grey,
  style=wireframe):
> plots[display](Plocha,Rov_z);
```




```
>
```